
Beispiel 18 Tellerfeder 

(Geometrische Nichtlinearität) 
 
 
Lernziel 
 
In diesem Beispiel wird eine nichtlineare strukturmechanische Berechnung vorge-
stellt. Es liegt eine geometrische Nichtlinearität vor, das Bauteilverhalten ist von 
den Verformungen abhängig. Die Ergebnisse der FEM-Berechnung werden mit 
Formeln verglichen. Zum Kennenlernen der Parametersprache APDL werden die 
Formeln mit der APDL (ANSYS parametric design language) programmiert und 
die Ergebnisse grafisch dargestellt. 
 
Es wird vorausgesetzt, dass Sie die Beispiele 1 und 2 durchgearbeitet haben und 
damit bereits das ANSYS/ED-Menüsystem und lineare strukturmechanische Berech-
nungen kennengelernt haben. 
 
Für die Durchführung von diesem Beispiel sehen Sie etwa 1 Stunde vor. 
 
 
1 Aufgabenstellung 
 
Die zu berechnende Struktur stellt eine Tellerfeder dar. Am inneren Rand wird eine 
axial wirkende Last aufgebracht. Der Außenrand ist axial gelagert. 
 
Reibungseinflüsse wer-
den an der Lasteinleitung 
und an der Auflagefläche 
sind geringfügig und wer-
den vernachlässigt. So-
wohl das Bauteil als auch 
die Lasten und Randbe-
dingungen sind rotations-
symmetrisch. 
 
Folgende Daten der Geometrie sind gegeben: 

Innendurchmesser  Di  = 141 mm 
Außendurchmesser  De = 198 mm 
Bauhöhe (unbelastet)  lo  = 8,15 mm 
Materialdicke  t = 2 mm 

 
Für das Material sind folgende Daten gegeben: 

Elastizitätsmodul:  E = 2,06 * 105 N/mm² 
Querkontraktionszahl  v = 0,3 

 
Gesucht wird die Kraft-Verformungs-Kurve der Tellerfeder. Dabei soll die Berech-
nung zunächst mit der Finite-Element-Methode mit dem ANSYS/ED-Programm erfol-
gen. Anschließend sollen die Berechnungsergebnisse mit den Werten verglichen 
werden, die sich aus den Formeln der technischen Berechnungsgrundlagen (DIN 
2092) ergeben. Die Verfasser bedanken sich bei der Fa. Christian Bauer, Welzheim 
für die freundliche Unterstützung und die zur Verfügung gestellten Unterlagen. 

 
Abb.1   Aufgabenstellung 

2 Idealisierung 
 
Die Aufgabe wird mit einer nichtlinearen statischen strukturmechanischen Berech-
nung gelöst. Das Bauteil wird durch ein 2-dimensionales rotationssymmetrisches 
Modell abgebildet. Dazu wird der Elementtyp PLANE42 verwendet, der in der Ebene 
definiert ist, die durch die radiale (x) und die axiale (y) Richtung gebildet wird. Jedes 
Element wird als Viereck mit 4 Eckknoten ausgeführt werden. Die Rotationssymme-
trie wird für diesen Elementtyp durch eine Schaltergröße (key option) festgelegt. Ein 
Element dieses Typs repräsentiert damit einen geschlossenen Ring um die Mittel-
achse. 
 
 
3 Das Preprocessing 
 
Der Aufruf des Preprocessors erfolgt mit 
/PREP7 
 
Der Elementtyp 
ET,1,PLANE42,,,1 
 
Die Materialdaten werden eingegeben mit 
MP,EX,1,2.06E5 
MP,NUXY,1,0.3 
und die bisherige Arbeit gesichert mit 
SAVE 
 
Die Ecken des Querschnitts der unbelasteten Tellerfeder sind durch den Innendurch-
messer, den Außendurchmesser, die Bauhöhe und die Materialdicke konstruktiv vor-
gegeben. Diese Abmessungen und einige weitere, daraus abzuleitende Abmessun-
gen werden von uns direkt bei der Festlegung dieser Ecken als arithmetische Aus-
drücke eingetragen. Das Programm führt die Berechnung jedes Zahlenwertes aus. 
K,1,(198/2)-0.5822 
K,2,198/2,1.9133 
K,3,(141/2)+0.5822,8.15 
K,4,141/2,8.15-1.9133 
 
Darstellung der Keypoints mit Nummern 
/PNUM,KP,1 
KPLOT 
 
Die Fläche 
A,1,2,3,4 
 
Darstellung der Fläche mit Linien-Nummern 
/PNUM,LINE,1 
APLOT 
 
Sichern Sie die bisherige Arbeit mit 
SAVE 
 
Die Vernetzungskriterien 
LESIZE,1,,,3 
LESIZE,2,,,15 

Die Vernetzung der Fläche 
AMESH,1 
 
Sichern Sie die bisherige Arbeit mit 
SAVE 
 
 
4 Die Lösung 
 
Der Aufruf des Lösungsteils 
/SOLU 
 
Die Randbedingung, die das Auflager darstellt 
DK,1,UY 
 
Die Last wird als aufgezwungene axiale Verschiebung des Eckpunktes (keypoint) 3 
eingegeben mit 
DK,3,UY,-12 
 
Die Tellerfeder hat dadurch, dass sie einen geschlossenen Kreisring darstellt, ein 
Tragverhalten, das nicht so leicht auf den ersten Blick zu durchschauen ist. Im 
vorliegenden Fall liegt eine aufgezwungene Verformung in axialer Richtung vor. Eine 
solche Verformung flacht zunächst die Tellerfeder ab. Dabei ist zu beachten, dass im 
Federquerschnitt sowohl Spannungen in der Ebene unseres Modells (also radiale 
und axiale Spannungen) als auch Umfangsspannungen (senkrecht auf die Ebene 
unseres Modells) auftreten. Diese Umfangsspannungen tragen erheblich zum 
gesamten Tragverhalten des Bauteils bei. 
 
Solange die Abflachung der Tellerfeder gering bleibt, liegt ein lineares Verhalten vor. 
Dann liegt ein linearer Zusammenhang zwischen der aufgeprägten Verschiebung, 
den Spannungen und der Reaktionskraft in der Tellerfeder vor. Bei größeren 
Verformungen ist dieser lineare Zusammenhang nicht mehr gegeben. Dann können 
bei der Tellerfeder die Spannungen und die Reaktionskraft langsamer zunehmen als 
die aufgeprägte Verschiebung. Nach einem Maximum kann sogar trotz zunehmen-
der Verschiebung eine Abnahme der Reaktionskraft auftreten! Dieser Zusammen-
hang zwischen dem Tragverhalten des Bauteils (also den Spannungen und Reak-
tionskräften) und den Verschiebungen ist damit nichtlinear, obwohl keine Plastizität 
des Materials und keine nichtlinearen äußeren Einflüsse vorliegen. Nur durch die 
Verformung ändert sich die Steifigkeit. Damit liegt eine Geometrie-Nichtlinearität 
vor. 
 
Im ANSYS-Programm wird standardmäßig eine lineare Geometrie angenommen. 
Wenn das Bauteil möglicherweise geometrisch nichtlinear ist, muss die Berück-
sichtigung dieses Effektes eingeschaltet werden 
NLGEOM,ON 
 
Bei einer nichtlinearen Berechnung kann das Ergebnis der Berechnung nicht in 
einem Berechnungsschritt erreicht werden. Ein Lösungsschritt bei der FEM besteht 
in der Lösung des linearen Gleichungssystems mit der Steifigkeitsmatrix und den 
unbekannten Zahlenwerten der Freiheitsgrade. Wenn das Bauteil nichtlinear ist, so 
a) muss ein solcher Lösungsschritt durchgeführt werden, 

b) anschließend müssen die berechneten Verschiebungen verwendet werden, 

um die Einflüsse der Bauteilverformung in der Steifigkeitsmatrix zu berück-
sichtigen, 

c) mit der neuen Steifigkeitsmatrix muss ein neuer Lösungsschritt durchgeführt 
werden, 

d) es muss geprüft werden, ob die neue Lösung zutreffend ist und abhängig 
davon 

e) muss entweder neu gerechnet werden (also wieder mit b weitermachen) oder 
die Folge von Lösungsschritten beendet werden. 

 
Jeder dieser Lösungsschritte wird als Gleichgewichtsiteration (equilibrium itera-
tion) bezeichnet. Wenn die Folge der Iterationen ein zutreffendes Ergebnis geliefert 
hat, wird davon gesprochen, dass die Berechnung konvergierte (converged). 
 
In vielen Fällen ist es ausreichend, den Schalter für die geometrische Nichtlinearität 
(large deform. effects) einzuschalten. Dadurch wird automatisch die Folge der 
Gleichgewichtsiterationen solange ausgeführt, bis die Berechnung konvergiert ist. 
Alle Standardeinstellungen (z.B. der Konvergenzkriterien) sind so gewählt, dass sie 
robust und alltagstauglich sind. Nur in besonderen Fällen sind die folgenden zusätz-
lichen Schritte in Betracht zu ziehen. 
 
Bei der Erläuterung der Gleichgewichtsiterationen war zugrundegelegt worden, dass 
iterativ das Ergebnis bestimmt wird, das für das nichtlineare Bauteil bei einer gege-
benen Last zutrifft. Es wird also mit der Dateneingabe eine bestimmte Last vorge-
geben und durch die Iterationen vom Programm das für diese Last zutreffende 
Ergebnis bestimmt. Bei nichtlinearen Berechnungen ist es aber häufig sinnvoll, 
Zwischenschritte zu untersuchen, zum Beispiel, 

• weil das Verhalten des Bauteils nicht nur im unbelasteten Zustand (am An-
fang) und bei voller Last (am Ende), sondern auch dazwischen untersucht 
werden soll und 

• weil bei nichtlinearen Bauteilen vielfach das Endergebnis auch vom Gesche-
hen dazwischen abhängig ist, also von der Folge der Schritte bis zu diesem 
Endergebnis. Diese Berechnungen sind nicht-konservativ, das Superposi-
tionsprinzip gilt dann nicht mehr. 

 
Insgesamt gibt es bei der Berechnung von nichtlinearen Bauteilen 

• Lastschritte (load steps): Jeder Lastschritt enthält eine Zusammenstellung 
von Lasten und Randbedingungen und Lastschrittoptionen zur Steuerung der 
Berechnung. Ein Lastschritt wird in der Dateneingabe mit einer Aufforderung 
an das Programm abgeschlossen, die Lösung und damit den Bauteilzustand 
am Ende dieses Lastschrittes (ausgehend vom Ende des vorangegangenen 
Lastschrittes) zu berechnen.  

• Zwischenschritte (substeps): Dies sind Bauteilzustände im Verlauf eines 
Lastschrittes, die ebenfalls berechnet werden sollen (zum Beispiel um den 
Verlauf genauer untersuchen zu können). In der Dateneingabe wird die 
Anzahl der Zwischenschritte (oder die Schrittweite) im Lastschritt angegeben. 

• Gleichgewichtsiterationen (equilibrium iterations): Für jeden 
Bauteilzustand muss eine Folge von linearen Gleichungslösungen ablaufen, 
bis das Gleichgewicht ausreichend genau erfüllt ist und Konvergenz erreicht 
ist. Jede dieser Gleichungslösungen ist eine Gleichgewichtsiteration. Bei den 
meisten Aufgaben sind die standardmäßig zugelassenen 25 Iterationen 
ausreichend, um Konvergenz zu erreichen. Die Standardeinstellung kann vom 
Anwender verändert werden (Kommando NEQIT). 

Im vorliegenden Beispiel ist ein einziger Lastfall vorgesehen. Es sollen aber 100 
Zwischenschritte (substeps) ausgeführt werden, ehe die aufgebrachte Last am Ende 
des Lastschrittes beim letzten Substep in voller Höhe berücksichtigt wird. 
 
Grundsätzlich gilt, dass eine solche nichtlineare Berechnung durch viele Zwischen-
schritte genauer wird, aber auch mehr Rechenzeit benötigt. Um dabei den optimalen 
Zwischenweg zu finden, kann eine Automatik verwendet werden, die laufend 
während des Berechnungsfortschrittes die Schrittweite zwischen den Zwischen-
schritten den Erfordernissen anpasst. Diese Automatik wird als Last- oder Zeit-
schrittautomatik (automatic time stepping) bezeichnet (sie ist nicht nur für nicht-
lineare statische Berechnungen, sondern insbesondere für dynamische Berech-
nungen zur Wahl des geeigneten Zeitschrittes (time step) wichtig). Die Aufgabe für 
den Anwender besteht nun nur noch darin, eine Anfangs-Schrittweite vorzugeben. 
Anschließend übernimmt die Automatik die Kontrolle der Schrittwahl innerhalb von 
vorgegebenen Grenzwerten. 
 
Im vorliegenden Beispiel wird die Last- oder Zeitschrittautomatik eingeschaltet. Am 
Anfang des Lastschrittes soll eine Schrittweite von 1/100 des Lastschrittes gewählt 
werden. Anschließend - also vom zweiten Zwischenschritt an - darf die Automatik die 
Schrittweite von einem Zwischenschritt zum nächsten dem Bauteilverhalten 
anpassen. Die Automatik soll aber die Grenzwerte einhalten, dass höchstens 1000 
Zwischenschritte und mindestens 30 Zwischenschritte ausgeführt werden. Diese 
Festlegungen werden eingegeben mit 
AUTOTS,ON 
NSUB,100,1000,30 
 
Standardmäßig werden nicht alle Ergebnisse der Substeps auf der Ergebnisdatei .rst 
abgespeichert, sondern nur die Ergebnisse des letzten Zwischenschrittes eines Last-
schrittes. Im vorliegenden Beispiel möchten wir jedoch das Ergebnis für jeden Zwi-
schenschritt abspeichern. Dies kann eingestellt werden mit 
OUTRES,ALL,ALL 
 
Um die Folge der Iterationen schneller auf die Konvergenz hinzuführen, können eine 
Anzahl von numerischen Hilfen eingeschaltet oder ausgeschaltet werden. Im vorlie-
genden Beispiel wird ein Prediktor (predictor) eingeschaltet 
PRED,ON 
 
Sichern Sie die bisherige Arbeit mit 
SAVE 
 
Die Lösung des Lastschrittes wird begonnen durch 
SOLVE 

Anschließend wird jeweils ein Zwischenschritt 
(substep) mit der ersten Iteration begonnen 
und je nach Erfordernis eine Folge von 
weiteren Gleichgewichtsiterationen ausge-
führt, bis Konvergenz erreicht wird (Verfolgen 
Sie die Berechnungslösung im Fenster 
ANSYS_OUTPUT. Beachten Sie die Meldun-
gen "SOLUTION CONVERGED".). 
 
Im Grafikfenster wird der Berechnungsfort-
schritt durch eine grafische Darstellung skiz-
ziert. Mit diesem „Monitor“ kann überwacht 
werden, welche Zahlenwerte aktuell für das 
Ungleichgewicht im berechneten Bauteil 
vorliegen (standardmäßig werden F L2, U L2 
verfolgt) und welche Zahlenwerte aktuell für 
das zugehörige Konvergenzkriterium (F 
CRIT, U CRIT) zugrunde gelegt werden. Immer wenn eine Iteration für beide Größen 
F und U feststellt, dass das Ungleichgewicht geringer als das Kriterium ist, wird 
diese Iteration als konvergiert festgestellt. Solange dies nicht der Fall ist, wird eine 
weitere Iteration gestartet. 
 
Die Kunst des Programmherstellers ist es zu erreichen, dass das Programm auto-
matisch die Annahmen für die neue Iteration so einstellt, dass die Ergebnisse dichter 
an das Ziel herankommen (und so das Ungleichgewicht geringer wird). Und dass 
dies möglichst effektiv in Hinsicht auf die benötigte Rechenzeit erfolgt. Im ANSYS-
Programm sind die zugehörigen Einstellungen so praxisgerecht wie möglich 
vorgesehen. Trotzdem bestehen viele Einwirkungsmöglichkeiten des Anwenders. 
Wenn in der Ingenieurs-Praxis ein nichtlineares Problem zu berechnen ist, sollten in 
Versuchsmodellen diese Einstellungen zunächst getestet werden. Dabei sollten die 
nichtlinearen Effekte in ihrem prinzipiellen Verhalten im Modell enthalten sein, die 
Modellgröße sollte aber so gering wie möglich sein. Im günstigsten Fall wird ein 
Modell mit einem einzigen Element aufgebaut, so dass selbst 1000 Iterationen in 
wenigen Sekunden abgearbeitet sind. Mit solchen Modellen können natürlich viele 
Varianten geprüft und verglichen werden, so dass der „ernste“ Rechenlauf gleich bei 
der ersten Anwendung effektiv, schnell und zielgerichtet abgeschlossen werden 
kann. 
 
 
5 Das Postprocessing 
 
Für die Darstellung der Ergebnisse wird der Time-History-Postprocessor POST26 
verwendet. Wenn auch hier die Berechnung keine Abhängigkeit von der Zeit enthielt, 
liegt doch durch den Berechnungsablauf als Folge von Substeps eine vergleichbare 
Datenmenge vor. So wie bei einer Zeitverlaufberechnung der Verlauf einzelner 
Ergebnisgrößen über der Folge von Zeitschritten ausgewertet wird, möchten wir den 
Verlauf einzelner Ergebnisgrößen über der Folge der Substeps auswerten. 
 
Der Aufruf des Time History Postprocessors erfolgt mit 
/POST26 
 
Mit dem folgenden Kommando wird die Nummer desjenigen Knotens abgefragt, der 
am Keypoint 3 (Ecke oben links) angeordnet wurde. Auf diese Weise kann bei der 

 
Abb.2   Lösungs-Verlauf, Monitor 

Eingabe mit Kommandos der interaktive Blick auf das Grafikfenster und das Ablesen 
der Nummer automatisiert werden 
NN=NODE(141/2+0.5822,8.15,0) 
 
Bei dem entsprechenden Menüschritt ist das Ablesen der Nummer erforderlich 
/PNUM,NODE,1 
NPLOT 
und die Nummer des Knotens oben links ablesen (bei dieser Vernetzung ist dies der 
Knoten 5). 
 
Die Verschiebung dieses Knotens in y-Richtung wird als eine der von uns auszu-
wertenden Ergebnisgrößen festgelegt. (Für die Kommando-Eingabe kann die 
Variable mit der Bezeichnung NN verwendet werden. Dieser Variablen wurde direkt 
vorher die Nummer des gewünschten Knotens zugewiesen. Für die Menü-Eingabe 
muss die Nummer des gewünschten Knotens aus dem Plotbild eingesetzt werden.) 
NSOL,2,NN,U,Y 

 
Durch diese Operation wurde die Folge derjenigen Zahlenwerte aus den 
Ergebnissen herausgesucht, die die y-Verschiebung des gewünschten Knotens aus 
jedem gerechneten Zwischenschritt (substep) darstellen. Diese Wertefolge (table) 
kann anschließend unter ihrer Referenznummer 2 weiterverwendet werden. 
 
Als Wertefolge (table) 1 werden automatisch die Zeitwerte jedes Substeps 
eingetragen. Wenn (wie in unserer Berechnung) die Zeit nicht direkt als Problemzeit 
vorgegeben war, wird automatisch das Ende des ersten Lastschrittes mit der Zeit t = 
1 versehen. Dementsprechend hat jeder Substep einen Zeitwert zwischen t = 0 
(Beginn der Berechnung) und t = 1 (letzter Substep). In diesem Fall kann am 
Zeitwert der Berechnungsfortgang (time, von 0 bis 1 zunehmend) abgelesen werden. 
 
Die Reaktionskraft in y-Richtung wird als zweite der von uns auszuwertenden 
Ergebnisgrößen für die Wertefolge (table) mit der Referenznummer 3 festgelegt mit 
RFOR,3,NN,F,Y 
 
Es soll ein Diagramm erstellt werden, in dem horizontal die y-Verschiebung und 
vertikal die y-Kraft aufgetragen werden. Beide Größen sind jedoch - von Null 
ausgehend - negativ. Daher wird für alle Einträge beider Tabellen 2 und 3 das 
Vorzeichen umgedreht. Diese arithmetische Operation wird hier auf verschiedene 
Weise durchgeführt, um die Möglichkeiten der Postprocessing-Operationen aufzu-
zeigen. 
 
Die Werte der Tabelle 2 (y-Verschiebung) werden jeweils mit dem Absolutwert 
überschrieben. Dadurch werden alle bisher negativen Werte positiv. Natürlich ist 
diese Operation nur dann eine Vorzeichen-Invertierung, wenn alle Zahlenwerte 
vorher negativ waren. Dies ist hier der Fall, daher kann diese Operation hier vor-
gestellt werden. 
ABS,2,2 
 
Die Werte der Tabelle 3 (y-Reaktionskraft) werden in einer Additions-Operation 
verwendet. Bei dieser Operation können mehrere Tabellen additiv überlagert 
werden, wobei vorher jeweils mit einem Faktor multipliziert wird. Die Operation kann 
beschrieben werden als 
 

IR = IA * FACTA + IB * FACTB + IC * FACTC 

IR Referenznummer der resultierenden Tabelle 
IA Referenznummer der ersten zu summierenden Tabelle 
FACTA Faktor für die Werte von Tabelle IA 
IB Referenznummer der zweiten zu summierenden Tabelle 
FACTB Faktor für die Werte von Tabelle IB 
IC Referenznummer der dritten zu summierenden Tabelle 
FACTC Faktor für die Werte von Tabelle IC 
 
Im vorliegenden Fall wird Tabelle 3 mit dem Faktor FACTA = -1 multipliziert und 
keine weiteren Werte hinzuaddiert. Das Ergebnis wird wieder in Tabelle 3 einge-
tragen 
ADD,3,3, , , , , ,-1 
 
Zur grafischen Darstellung wird festgelegt, dass Tabelle 2 entlang der horizontalen 
Achse des Diagrammes aufzutragen ist. 
XVAR,2 
 
In Vertikalrichtung sollen die Zahlen-
werte von Tabelle 3 aufgetragen werden. 
PLVAR,3 
 
Das Diagramm wird zusätzlich noch mit 
einer Achsenbeschriftung und einem 
bestimmten Wertebereich auf den 
Achsen festgelegt mit 
/GRID,1 
/AXLA,X,Durchbiegung Innenrand 
(mm) 
/AXLA,Y,Kraft F (N) 
/XRAN,0,12 
/YRAN,-1000,6000 
 
Mit diesen Spezifikationen wird das Diagramm wiederholt. 
PLVAR,3 
 
Die Berechnung ist damit abgeschlossen, die ANSYS-Programmanwendung kann 
abgeschlossen werden mit 
FINISH 
/EXIT 
 
 
6 Varianten zum Selbststudium 
 
Zum Studium der ANSYS-Parametersprache (APDL - ANSYS parametric design 
language) kann die analytische Lösung programmiert werden und mit der FEM-
Lösung verglichen werden. Die Formeln der analytischen Lösung entsprechend DIN 
2092 sind hier dargestellt (aus “Tellerfeder”, Technische Datensammlung, Christian 
Bauer GmbH&Co, Welzheim). 
 
1. Kennwerte 

i

e

D
D

=δ  

tlh −= 00  

  

 
Abb.3   Kraft-Weg-Diagramm 

Leseprobe © CADFEM GmbH 2013



Beispiele   Teil IV   Beispiel 18  695 
 

δδ
δ

δ
δ

π
ln

2
1
1

1
1

2

1

−
−
+







 −

⋅=K  

δ
δ

δ

π ln

1
ln

1
6

2

−
−

⋅=K  

δ
δ

π ln
13

3
−

⋅=K  

 
Für Tellerfedern ohne Auflagefläche gilt 

14 =K  
 
 
2. Federkraft 

3. Bezeichnungen 

μ ....  Querkontraktionszahl ( = v) 
s ....  Federweg der Einzeltellerfeder 
h0 .... rechnerische Hilfsgröße: h0 = l0 - t 

(für Tellerfeder ohne Auflagefläche) 
F .... Kraft der Einzeltellerfeder 
Fc ....  rechnerische Kraft der Einzeltellerfeder in Planlage 

 
Die Formeln lassen es zu, den Zusammenhang zwischen der y-Verschiebung und 
der y-Kraft für jeden Lastpunkt zu berechnen. In den folgenden Eingabe-
Kommandos werden diese Formeln einprogrammiert. In der APDL werden dabei 
arithmetische Operationen ausgeführt. Es werden sowohl skalare Variablen verwen-
det als auch Feldvariablen (indizierte Variablen, vector, array) dimensioniert und 
verwendet. Diese Eingabe sollte immer als Kommandofolge in einem Editor des 
Rechner-Betriebssystems geschrieben und durch ein Lese-Kommando (Kommando 
/inp, Menü-Aufruf Utility Menu> File> Read Input From) im ANSYS-Programm 
ausgeführt werden. Nur in den seltensten Fällen ist hierfür die direkte Eingabe 
(Utility Menu> Parameters) sinnvoll. 
 
Hier wird empfohlen, die folgenden Zeilen im Editor zu schreiben und zum Beispiel 
als Datei "ANALY.EIN" abzuspeichern. 
 
/TITLE, TELLERFEDER 
EX=2.06E5  ! Parameter definieren, zur Darstellung 
NU=0.3  ! der analytischen Lösung 
DE=198 
DI=141 
T=2 
L0=8.15 

PI=3.14159 
DELT=DE/DI 
H0=L0-T 
T1=((DELT-1)/DELT)**2 
T2=(DELT+1)/(DELT-1) 
T3=2/LOG(DELT) 
K1=T1/(PI*(T2-T3)) 
C12=(4*EX)/(1-NU**2)*(T**4/(K1*DE**2)) 
*DIM,S,TABLE,100 
*DIM,F,TABLE,100 
*VFILL,S(1,0),RAMP,1,1 
*VFILL,S(1),RAMP,.12,.12 
*VFILL,F(1,0),RAMP,1,1 
*DO,I,1,100,1 
D1=S(I)/T 
D2=(H0/T-S(I)/T)*(H0/T-S(I)/(2*T))+1 
F(I)=C12*D1*D2 
*ENDDO 
U=12 
A=4 ! Keypoint Koordinaten (x,y) 
B=-4*L0 
C=L0**2-4*T**2+(DE/2-DI/2)**2 ! FX = AX4 + BX3 + CX² + DX + E 
D=2*L0*T**3 
E=T**4-(DE/2-DI/2)**2*(T**2) ! Newton Verfahren 
XI=T 
*DO,I,1,10,1 
FX=A*XI**4+B*XI**3   ! FXP = ¶FX/¶X 
FX=FX+C*XI**2+D*XI+E 
FXP=4*A*XI**3+3*B*XI**2  ! X(i+1) = X(i) - FX/FXP 
FXP=FXP+2*C*XI+D 
X=XI-FX/FXP 
XX=ABS(X-XI) 
XI=X 
*ENDDO 
Y=SQRT(T**2-X**2) 
 
Wenn diese Datei vorhanden ist, kann im Postprocessing nach dem Erstellen des 
Diagramms der y-Kraft, aufgetragen über der y-Verschiebung, die analytische 
Lösung zum Vergleich aufgetragen werden. 
 
Die analytische Lösung wird ausgeführt mit 
/INP,ANALY,EIN 
 
Das vorhandene Diagramm soll im Grafikfenster erhalten bleiben, wenn die neue 
Kurve darüber gezeichnet wird 
/NOER 
 
Die Array-Zahlenwerte werden als Diagramm gezeichnet 
*VPLO,S(1),F(1) 

